

Example 7d: Explanatory IRT Models as Crossed Random Effects Models in SAS GLIMMIX

This example demonstrates variants “explanatory” item response theory (IRT) models, which are actually generalized multilevel models with random subject intercepts and either fixed or random effects for items. These example data are from my dissertation: 36 items assessing attentional search via change detection scored incorrect (cor=0) or correct (cor=1). Items varied by four dimensions: continuous visual clutter (Rclut), whether the change was relevant to driving (Rrel), continuous brightness of the change (Rbrit), and whether the change was made to a legible sign (LegSign).

These analyses require a “stacked” (or “long”) format in which each item for each person is stored on a separate row.

SAS Code to read in already-stacked data:

```
* Reading in data and keeping items with item predictor variables;
DATA work.o38stack; SET diss.work.o38stack; WHERE NMISS(Rclut, Rrel, Rbrit, LegSign)=0; RUN;
PROC SORT DATA=work.o38stack; BY PartID picture; RUN;
```

Empty Model with Random Subjects Only via GLIMMIX: $\text{Logit}(y_{is}=1|\theta_s) = \gamma_{00} + \theta_s$

```
TITLE1 "Random Subjects, Fixed Items Models for Correct Responses";
TITLE2 "Empty Model (Exchangeable Items)";
PROC GLIMMIX DATA=work.o38stack NOCLPRINT NAMELEN=100 METHOD=Laplace GRADIENT;
  CLASS PartID picture;
  MODEL cor (DESCENDING) = / SOLUTION LINK=LOGIT DIST=BINARY; * Binary DV, logit link;
  RANDOM INTERCEPT / SOLUTION TYPE=UN SUBJECT=PartID; * Each subject gets a theta(not shown);
RUN;
```

The GLIMMIX Procedure

Model Information	
Data Set	WORK.WORK.O38STACK
Response Variable	COR
Response Distribution	Binary
Link Function	Logit
Variance Function	Default
Variance Matrix Blocked By	PARTID
Estimation Technique	Maximum Likelihood
Likelihood Approximation	Laplace
Degrees of Freedom Method	Containment

Convergence criterion (GCONV=1E-8) satisfied.

Fit Statistics

-2 Log Likelihood	5600.45	LL*-2 already, and this is baseline model for assessing fit
AIC (smaller is better)	5604.45	
AICC (smaller is better)	5604.45	
BIC (smaller is better)	5610.54	
CAIC (smaller is better)	5612.54	
HQIC (smaller is better)	5606.92	

Covariance Parameter Estimates

Cov	Subject	Estimate	Standard Error	Gradient
UN(1,1)	PARTID	0.2524	0.05033	-0.00018 → variance of theta (mean = 0): persons are random

Solutions for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > t	Gradient
Intercept	1.3487	0.05367	154	25.13	<.0001	0.000155 → item mean easiness in logits

Explanatory Items Model with Random Subjects Only (Predicted Item Easiness) via GLIMMIX:

$$\text{Logit}(y_{is}=1|\theta_s) = \gamma_{00} + \gamma_{10}\text{Rclut}_i + \gamma_{20}\text{Rrel}_i + \gamma_{30}\text{Rbrit}_i + \gamma_{40}\text{LegSign}_i + \theta_s$$

```
TITLE1 "Random Subjects, Fixed Items Models for Correct Responses";
TITLE2 "LLTM Model (Easiness-Predicted Items)";
PROC GLIMMIX DATA=work.o38stack NOCLPRINT NAMELEN=100 METHOD=Laplace GRADIENT;
CLASS PartID picture;
MODEL cor (DESCENDING) = Rclut Rrel Rbrit LegSign / SOLUTION LINK=LOGIT DIST=BINARY;
RANDOM INTERCEPT / SOLUTION TYPE=UN SUBJECT=PartID; RUN;
```

Convergence criterion (GCONV=1E-8) satisfied.

Fit Statistics

-2 Log Likelihood	5439.80	→ -2LL difference (df=4) relative to empty model is significant
AIC (smaller is better)	5451.80	
AICC (smaller is better)	5451.81	
BIC (smaller is better)	5470.06	
CAIC (smaller is better)	5476.06	
HQIC (smaller is better)	5459.21	

Covariance Parameter Estimates

Cov	Subject	Estimate	Standard Error	Gradient
UN(1,1)	PARTID	0.2734	0.05360	-0.0000 → theta variance (note is bigger now)

Solutions for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > t	Gradient
Intercept	0.8619	0.1529	154	5.64	<.0001	-0.00104 → mean logit easiness if x=0
RCLUT	-0.2675	0.05548	5267	-4.82	<.0001	-0.00001 → Δ in logit/unit clutter
RREL	0.2204	0.09936	5267	2.22	0.0266	-0.00149 → Δ in logit if relevant
RBRIT	0.4742	0.1129	5267	4.20	<.0001	-0.00061 → Δ in logit/unit brightness
LEGSIGN	0.6621	0.08223	5267	8.05	<.0001	9.234E-6 → Δ in logit if legible sign

Rasch Model with Random Subjects and Fixed Items (via dummy codes) via GLIMMIX:

$$\text{Logit}(y_{is}=1|\theta_s) = \gamma_{00} + \gamma_{10}\text{Pic2}_i + \gamma_{20}\text{Pic3}_i + \dots + \gamma_{350}\text{Pic35}_i + \theta_s$$

```
TITLE1 "Random Subjects, Fixed Items Models for Correct Responses";
TITLE2 "Rasch Model (Fixed Items, Easiness for Each)";
PROC GLIMMIX DATA=work.o38stack NOCLPRINT NAMELEN=100 METHOD=Laplace GRADIENT;
CLASS PartID picture;
MODEL cor (DESCENDING) = picture / SOLUTION LINK=LOGIT DIST=BINARY;
RANDOM INTERCEPT / SOLUTION TYPE=UN SUBJECT=PartID;
LSMEANS picture; ODS OUTPUT LSMEANS=work.Rasch; RUN;
```

Convergence criterion (GCONV=1E-8) satisfied.

Fit Statistics

-2 Log Likelihood	4907.49	→ -2LL difference (df=35) relative to empty model is significant
AIC (smaller is better)	4981.49	
AICC (smaller is better)	4982.01	
BIC (smaller is better)	5094.09	
CAIC (smaller is better)	5131.09	
HQIC (smaller is better)	5027.22	

Covariance Parameter Estimates

Cov	Subject	Estimate	Standard Error	Gradient
UN(1,1)	PARTID	0.3678	0.06794	-0.00004 → theta variance (biggest yet)

Type III Tests of Fixed Effects

Effect	Num	Den	F Value	Pr > F
PICTURE	35	5236	15.41	<.0001 → items do vary significantly from each other in easiness

PICTURE Least Squares Means → easiness fixed effect estimates per item (saved to dataset)

Picture Number	Estimate	Standard Error	DF	t Value	Pr > t
2	0.2306	0.1781	5236	1.29	0.1957
6	3.3655	0.4235	5236	7.95	<.0001
7	1.1464	0.1989	5236	5.76	<.0001
10	2.2291	0.2673	5236	8.34	<.0001
11	0.9651	0.1920	5236	5.03	<.0001
13	0.1150	0.1764	5236	0.65	0.5145
22	1.2780	0.2060	5236	6.21	<.0001
23	2.9246	0.3515	5236	8.32	<.0001
26	1.0325	0.1947	5236	5.30	<.0001
33	1.6398	0.2243	5236	7.31	<.0001
35	1.3106	0.2078	5236	6.31	<.0001
42	0.9466	0.1929	5236	4.91	<.0001
52	2.7153	0.3219	5236	8.44	<.0001
59	2.2571	0.2746	5236	8.22	<.0001
61	0.8884	0.1927	5236	4.61	<.0001
62	-0.3283	0.1801	5236	-1.82	0.0684
66	2.5329	0.2993	5236	8.46	<.0001
97	1.3749	0.2090	5236	6.58	<.0001
117	3.0615	0.3706	5236	8.26	<.0001
123	2.3657	0.2817	5236	8.40	<.0001
128	1.1477	0.1989	5236	5.77	<.0001
135	3.3718	0.4235	5236	7.96	<.0001
136	1.2737	0.2092	5236	6.09	<.0001
137	1.4945	0.2161	5236	6.92	<.0001
140	0.3818	0.1809	5236	2.11	0.0348
146	1.7911	0.2338	5236	7.66	<.0001
152	0.7385	0.1874	5236	3.94	<.0001
155	-0.2088	0.1783	5236	-1.17	0.2416
161	2.9072	0.3519	5236	8.26	<.0001
162	3.0362	0.3709	5236	8.19	<.0001
171	1.5756	0.2210	5236	7.13	<.0001
172	0.8620	0.1910	5236	4.51	<.0001
173	3.2169	0.3938	5236	8.17	<.0001
174	1.6938	0.2272	5236	7.45	<.0001
177	0.8978	0.1897	5236	4.73	<.0001
179	1.7847	0.2340	5236	7.63	<.0001

Empty Model with Random Subjects AND Random Items via GLIMMIX:

$$\text{Logit}(y_{is}=1|\theta_s) = \gamma_{00} + \theta_s + b_i$$

```

TITLE1 "Random Subjects, Random Items Models for Correct Responses";
TITLE2 "Empty Model (Exchangeable Items)";
PROC GLIMMIX DATA=work.o38stack NOCLPRINT NAMELEN=100 METHOD=Laplace GRADIENT;
  CLASS PartID picture;
  MODEL cor (DESCENDING) = / SOLUTION LINK=LOGIT DIST=BINARY;
  RANDOM INTERCEPT / SOLUTION TYPE=UN SUBJECT=PartID; * Each subject gets a theta;
  RANDOM INTERCEPT / SOLUTION TYPE=UN SUBJECT=picture; * Each item gets an easiness;
  ODS OUTPUT SolutionR=work.Crossed; RUN;

* Creating analogous individual item effects to fixed effects items model;
DATA work.Crossed; SET work.Crossed; WHERE INDEX(Subject, "PICTURE")>0;
  Random_b=1.5928 + Estimate; * Add fixed intercept back into each item random effect;
  KEEP Random_b; RUN;

```

Fit Statistics

-2 Log Likelihood	5049.84
AIC (smaller is better)	5055.84
AICC (smaller is better)	5055.84
BIC (smaller is better)	5049.84
CAIC (smaller is better)	5052.84
HQIC (smaller is better)	5049.84

Covariance Parameter Estimates

Cov		Estimate	Standard Error	Gradient
Parm	Subject			
UN(1,1)	PARTID	0.3625	0.06733	0.004887 → theta variance
UN(1,1)	PICTURE	0.9492	0.2431	-0.00131 → item easiness variance

Solutions for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > t	Gradient
Intercept	1.5928	0.1750	35	-9.10	<.0001	0.000010 → mean item easiness in logits

Solution for Random Effects → deviation of each item from 'mean' easiness (saved to dataset)

Effect	Subject	Estimate	Pred	DF	t Value	Pr > t
Intercept	PICTURE 2	1.3266	0.2337	5236	5.68	<.0001
Intercept	PICTURE 6	-1.4697	0.3819	5236	-3.85	0.0001
Intercept	PICTURE 7	0.4583	0.2475	5236	1.85	0.0641
Intercept	PICTURE 10	-0.5488	0.2935	5236	-1.87	0.0616
Intercept	PICTURE 11	0.6304	0.2430	5236	2.59	0.0095
Intercept	PICTURE 13	1.4364	0.2326	5236	6.18	<.0001
Intercept	PICTURE 22	0.3336	0.2524	5236	1.32	0.1863
Intercept	PICTURE 23	-1.1401	0.3452	5236	-3.30	0.0010
Intercept	PICTURE 26	0.5663	0.2448	5236	2.31	0.0207
Intercept	PICTURE 33	-0.00675	0.2648	5236	-0.03	0.9797
Intercept	PICTURE 35	0.3027	0.2537	5236	1.19	0.2329
Intercept	PICTURE 42	0.6476	0.2436	5236	2.66	0.0079
Intercept	PICTURE 52	-0.9695	0.3278	5236	-2.96	0.0031
Intercept	PICTURE 59	-0.5725	0.2985	5236	-1.92	0.0552
Intercept	PICTURE 61	0.7024	0.2434	5236	2.89	0.0039
Intercept	PICTURE 62	1.8513	0.2350	5236	7.88	<.0001
Intercept	PICTURE 66	-0.8153	0.3140	5236	-2.60	0.0094
Intercept	PICTURE 97	0.2422	0.2545	5236	0.95	0.3413
Intercept	PICTURE 117	-1.2488	0.3555	5236	-3.51	0.0004
Intercept	PICTURE 123	-0.6698	0.3029	5236	-2.21	0.0270
Intercept	PICTURE 128	0.4571	0.2475	5236	1.85	0.0648
Intercept	PICTURE 135	-1.4751	0.3818	5236	-3.86	0.0001
Intercept	PICTURE 136	0.3371	0.2546	5236	1.32	0.1856
Intercept	PICTURE 137	0.1295	0.2593	5236	0.50	0.6175
Intercept	PICTURE 140	1.1831	0.2356	5236	5.02	<.0001
Intercept	PICTURE 146	-0.1478	0.2713	5236	-0.54	0.5859
Intercept	PICTURE 152	0.8449	0.2398	5236	3.52	0.0004
Intercept	PICTURE 155	1.7400	0.2339	5236	7.44	<.0001
Intercept	PICTURE 161	-1.1244	0.3456	5236	-3.25	0.0011
Intercept	PICTURE 162	-1.2262	0.3560	5236	-3.44	0.0006
Intercept	PICTURE 171	0.05319	0.2627	5236	0.20	0.8395
Intercept	PICTURE 172	0.7276	0.2423	5236	3.00	0.0027
Intercept	PICTURE 173	-1.3674	0.3675	5236	-3.72	0.0002
Intercept	PICTURE 174	-0.05721	0.2668	5236	-0.21	0.8302
Intercept	PICTURE 177	0.6943	0.2414	5236	2.88	0.0040
Intercept	PICTURE 179	-0.1423	0.2714	5236	-0.52	0.6002

Explanatory Items Model with Random Subjects AND Random Items (Predicted Item Easiness)

$$\text{Logit}(y_{is}=1|\theta_s) = \gamma_{00} + \gamma_{10}Rclut_i + \gamma_{20}Rrel_i + \gamma_{30}Rbrit_i + \gamma_{40}LegSign_i + \theta_s + b_i$$

```
TITLE1 "Random Subjects, Random Items Models for Correct Responses";
TITLE2 "LLTM Model (Easiness-Predicted Items)";
PROC GLIMMIX DATA=work.o38stack NOCLPRINT NAMELEN=100 METHOD=Laplace GRADIENT;
  CLASS PartID picture;
  MODEL cor = Rclut Rrel Rbrit LegSign / SOLUTION LINK=LOGIT DIST=BINARY;
  RANDOM INTERCEPT / SOLUTION TYPE=UN SUBJECT=PartID; * Each subject gets a theta;
  RANDOM INTERCEPT / SOLUTION TYPE=UN SUBJECT=picture; * Each item gets an easiness;
RUN;
```

Fit Statistics

-2 Log Likelihood	5040.20
AIC (smaller is better)	5054.20
AICC (smaller is better)	5054.22
BIC (smaller is better)	5040.20
CAIC (smaller is better)	5047.20
HQIC (smaller is better)	5040.20

Covariance Parameter Estimates

Cov	Subject	Estimate	Standard Error	Gradient
UN(1,1)	PARTID	0.3625	0.06732	0.002419 → theta variance across subjects
UN(1,1)	PICTURE	0.7138	0.1858	0.000701 → remaining variance in item easiness

Solutions for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > t	Gradient
Intercept	1.3105	0.6354	32	2.06	0.0474	0.002994 → mean easiness when x=0
RCLUT	-0.3239	0.2420	5235	-1.34	0.1809	0.000473 → Δ in logit/unit clutter
RREL	0.03715	0.4261	5235	0.09	0.9305	0.004363 → Δ in logit if relevant
RBKIT	0.7896	0.4989	5235	1.58	0.1136	4.913E-6 → Δ in logit/unit brightness
LEGSIGN	0.7388	0.3367	5235	2.19	0.0283	0.000287 → Δ in logit if legible sign

Comparing Results: Fixed vs. Random Effects for Item Easiness Predictions

```
DATA work.items; SET work.o38stack; WHERE PartID=201; RUN;
DATA work.Merged; MERGE work.Rasch (RENAME=(Estimate=Fixed_b)) work.Crossed work.items; RUN;
DATA work.Merged; SET Merged;
LLTM_pred_b = 0.8619 + (-.2675*rclut) + (.22040*rrel) + (.4742*rbrit) + (.6621*legsign);
Crossed_pred_b = 1.3105 + (-.3239*rclut) + (.03715*rrel) + (.7896*rbrit) + (.7388*legsign);
RUN;
PROC CORR DATA=work.Merged; VAR fixed_b random_b LLTM_pred_b Crossed_pred_b; RUN;
PROC GPLOT DATA=work.Merged; PLOT fixed_b*random_b; RUN; QUIT;
```

Simple Statistics

Variable	N	Mean	Std Dev	Sum	Minimum	Maximum
Fixed_b	36	1.61155	1.02674	58.01583	-0.32828	3.37183
Random_b	36	1.63255	0.93089	58.77182	0.11072	3.43713
LLTM_pred_b	36	1.40093	0.44574	50.43335	0.60327	2.28516
Crossed_pred_b	36	1.59253	0.49180	57.33107	0.53570	2.58009

Pearson Correlation Coefficients, N = 36; Prob > |r| under H0: Rho=0

	Fixed_b	Random_b	LLTM_pred_b	Crossed_pred_b
Fixed_b	1.00000	0.99950	0.47628	0.48551
Random_b	0.99950	1.00000	0.47770	0.48561
LLTM_pred_b	<.0001	<.0001	1.00000	0.98472
Crossed_pred_b	0.48551	0.48561	0.98472	1.00000
	0.0027	0.0027	<.0001	

The R^2 for item easiness is around .24, which is very close to what we found as the proportion reduction in item variance after including the 4 item predictors.

The difference lies in the significance of the item predictor effects, which are based on the wrong df and error term in the random subjects only LLTM.